

Ткаченко Кирилл Станиславович

*Инженер 1-й категории, ассистент ФГАОУ ВО
«Севастопольский государственный университет».*



«Муравей Лэнгтона» на школьном алгоритмическом языке

В публикации описывается один из наиболее простых клеточных автоматов – «Муравей Лэнгтона». Программная реализация на школьном алгоритмическом языке симулирует функционирование этого автомата в течение конечного числа итераций. Приведённые полный исходный текст и результаты работы программы позволят всем изучающим и интересующимся школьным алгоритмическим языком воспроизвести и реализовать конечный автомат.

Многие задачи прикладного программирования имеют решение с использованием клеточных автоматов или подобных им методов и средств. Поэтому целесообразно изучать их, начиная как можно раньше, в том числе со школьной скамьи. При этом чем проще клеточный автомат, тем возможно проще его программная реализация. Одним из наиболее простых клеточных автоматов с довольно простыми правилами является так называемый «Муравей Лэнгтона». По этой

причине его можно рассмотреть и программно реализовать средствами школьного алгоритмического языка.

Целью настоящей публикации является разработка программы на школьном алгоритмическом языке, исполняющей правила клеточного автомата типа «Муравей Лэнгтона».

Правила работы автомата следующие. На прямоугольном, разбитом на ячейки поле находится муравей. Каждая ячейка может иметь либо белый, либо черный цвет, по умолча-

нию цвет белый. Муравей имеет положение на этой доске и направление перемещения. Направление перемещения – это либо «вверх» (изначально), либо «вправо», либо «вниз», либо «влево». Пока муравей находится на поле и может ещё совершать перемещения, он работает с ячейкой поля, где находится. Если цвет ячейки белый, то цвет меняется на чёрный и муравей поворачивается по часовой стрелке один раз. Иначе, если цвет ячейки чёрный, то цвет меняется на белый и муравей поворачивается против часовой стрелки один раз.

Для удобства направление «вверх» обозначается численно 0, «вправо» – 1, «вниз» – 2, «влево» – 3. Вначале определяются константы. Целочисленные константы **ШИРИНА** и **ВЫСОТА** хранят соответственно ширину и высоту поля в ячейках. Для наглядности алфавитно-цифрового вывода ширина поля и высота поля ограничиваются 64 ячейками. Ограничение числа итераций, равное 20 000, задаётся в целочисленной константе **НАИБ_ШАГ**. Символы, обозначающие цвет ячейки поля, помещаются в **ЦВЕТ_БЕЛЫЙ**

для белого цвета, и **ЦВЕТ_ЧЕРНЫЙ** для чёрного.

```
цел ШИРИНА = 64;  
цел ВЫСОТА = 64;  
цел НАИБ_ШАГ = 20000;  
сим ЦВЕТ_БЕЛЫЙ = ".";  
сим ЦВЕТ_ЧЕРНЫЙ = "0";
```

Вся программа помещается в рамках одного алгоритма-процедуры **Потенциал_Муравей_Лэнгтона**.

В начале этого алгоритма определяются переменные, в которых располагаются необходимые для функционирования программы данные. Целочисленные переменные **x** и **y** хранят положение муравья, **y** – строку массива поля, **x** – столбец, **Направление** – номер направления перемещения муравья, **Шаг** – номер шага, **i** и **j** – счётчики для циклов с параметром. Литерная переменная **Рез** накапливает в себе пользовательский вывод, а именно ASCII-картинку с результатом работы автомата. Двумерный массив **Поле** предназначен для представления информации о поле для автомата в программе.

```
алг Потенциал_Муравей_Лэнгтона  
нач  
цел x, y;  
цел Направление;  
цел Шаг;  
сим таб Поле[0 : ВЫСОТА - 1, 0 : ШИРИНА - 1];  
лит Рез;  
цел i, j;
```

Полагается, что муравей вначале находится в центре поля:

```
x := div(ШИРИНА, 2);  
y := div(ВЫСОТА, 2);
```

Вначале направление «вверх», а счётчик шагов равен нулю:

```
Направление := 0;  
Шаг := 0;
```

Всем ячейкам поля присваивается белый цвет:

```
нц для i от 0 до ВЫСОТА - 1
  нц для j от 0 до ШИРИНА - 1
    Поле[i, j] := ЦВЕТ_БЕЛЫЙ;
  кц
кц
```

Пока не выполнено ограничение на число шагов и муравей не покинул поле:

```
нц пока (Шаг < НАИБ_ШАГ) и (0 <= y) и (y < ВЫСОТА) и (0 <= x)
и (x < ШИРИНА)
```

Если цвет ячейки белый, то цвет муравья меняется на чёрный и муравей поворачивается по часовой стрелке один раз:

```
если Поле[y, x] = ЦВЕТ_БЕЛЫЙ
то
  Поле[y, x] := ЦВЕТ_ЧЕРНЫЙ;
  Направление := Направление + 1;
```

Иначе, если цвет ячейки чёрный, то цвет муравья меняется на белый и муравей поворачивается против часовой стрелки один раз:

```
иначе
  Поле[y, x] := ЦВЕТ_БЕЛЫЙ;
  Направление := Направление + 3;
все
```

Направление ограничивается четырьмя вариантами:

```
Направление := mod(Направление, 4);
```

Если направление «вверх», то координата строки уменьшается на единицу:

```
если Направление = 0
то
  y := y - 1;
```

Иначе если направление «вправо», то координата столбца увеличивается на единицу:

```
иначе если Направление = 1
то
  x := x + 1;
```

Иначе если направление «вниз», то координата строки увеличивается на единицу:

```
        иначе если Направление = 2
            то
                у := у + 1;
```

Иначе если направление «влево», то координата столбца уменьшается на единицу:

```
                иначе
                    х := х - 1;
            все
        все
    все
```

Увеличивается счётчик шагов:

```
        Шаг := Шаг + 1;
    кц
```

Для построения ASCII-картинки с результатом функционирования автомата очищается результирующая строка, для всех строк поля, для

всех столбцов поля содержимое ячейки конкатенируется с результатом, в конце каждой строки добавляется терминатор:

```
Рез := "";
нц для i от 0 до ВЫСОТА - 1
    нц для j от 0 до ШИРИНА - 1
        Рез := Рез + Поле[i, j];
    кц
    Рез := Рез + нс;
кц
```

Результат выводится на стандартное устройство вывода:

```
        вывод Рез, нс;
    кон
```

Полный текст программы приводится в приложении А, результаты работы – в приложении Б.

Вычислительные эксперименты с этой программой, в том числе изме-

нение правил функционирования автомата, позволят глубоко и в полной мере изучить работу с циклическими и ветвящимися структурами, а также обработку двумерных массивов и строк.

Приложение А. Полный текст программы

```
цел ШИРИНА = 64;  
цел ВЫСОТА = 64;  
цел НАИБ_ШАГ = 20000;  
сим ЦВЕТ_БЕЛЫЙ = ".";  
сим ЦВЕТ_ЧЕРНЫЙ = "o";  
алг Потенциал_Муравей_Лэнгтона  
нач  
    цел x, y;  
    цел Направление;  
    цел Шаг;  
    сим таб Поле[0 : ВЫСОТА - 1, 0 : ШИРИНА - 1];  
    лит Рез;  
    цел i, j;  
  
    x := div(ШИРИНА, 2);  
    y := div(ВЫСОТА, 2);  
    Направление := 0;  
    Шаг := 0;  
    нц для i от 0 до ВЫСОТА - 1  
        нц для j от 0 до ШИРИНА - 1  
            Поле[i, j] := ЦВЕТ_БЕЛЫЙ;  
        кц  
    кц  
  
    нц пока (Шаг < НАИБ_ШАГ) и (0 <= y) и (y < ВЫСОТА) и (0 <= x)  
и (x < ШИРИНА)  
        если Поле[y, x] = ЦВЕТ_БЕЛЫЙ  
            то  
                Поле[y, x] := ЦВЕТ_ЧЕРНЫЙ;  
                Направление := Направление + 1;  
            иначе  
                Поле[y, x] := ЦВЕТ_БЕЛЫЙ;  
                Направление := Направление + 3;  
        все  
        Направление := mod(Направление, 4);  
        если Направление = 0  
            то  
                y := y - 1;  
            иначе если Направление = 1  
                то  
                    x := x + 1;  
                иначе если Направление = 2  
                    то  
                        y := y + 1;  
                    иначе  
                        x := x - 1;  
                все  
            все  
        все  
    все
```

```

    все
      Шаг := Шаг + 1;
    кц

    Рез := "";

    нц для i от 0 до ВЫСОТА - 1
      нц для j от 0 до ШИРИНА - 1
        Рез := Рез + Поле[i, j];
      кц
      Рез := Рез + нс;
    кц

    вывод Рез, нс;
  КОН

```

Приложение Б. Результаты работы программы

[illegible]

Список литературы

1. Муравей Лэнгтона // Википедия.
https://ru.wikipedia.org/wiki/Муравей_Лэнгтона.
2. Langton's ant // Rosetta Code. http://rosettacode.org/wiki/Langton%27s_ant.
3. Ткаченко К. С. «Муравей Лэнгтона» в 1С// Системный администратор. Вып. 07–08 (188–189)/2018. – Москва: ООО «Издательский дом Положевец и партнеры», 2018. – С. 76–79.